Hungry Mice

When laboratory mice (and maybe other animals) are fed a nutritionally adequate but near-starvation diet, they may live longer on average than mice that eat a normal amount of food. In this experiment, mice were randomly assigned to one of six experimental treatments. The dependent variable was how long they lived before dying of old age. The experimental treatments were:

> **NP**: Mice in this group ate as much as they pleased of a non-purified, standard diet for laboratory mice.
>
> **N/N85**: This group was fed normally both before and after weaning. (The slash distinguishes the two periods.) After weaning, the ration was controlled at 85 kcal/wk. This, rather than NP, serves as the control group because caloric intake is held reasonably constant.
>
> **N/R50**: This group was fed a normal diet before weaning and a reduced-calorie diet of 50 kcal/wk after weaning.
>
> **R/R50**: This group was fed a reduced-calorie diet of 50 kcal/wk both before and after weaning.
>
> **N/R50 lopro**: This group was fed a normal diet before weaning, a restricted diet of 50 kcal/wk after weaning, and had dietary protein content decreased with advancing age.
>
> **N/R40**: This group was fed normally before weaning and was given a severely reduced diet of 40 kcal/wk after weaning.

The investigators were not interested in every possible comparison between treatment means. They had *very* specific research questions.

NP	N/N85	N/R50	R/R50	N/R50 lopro	N/R40
μ_1	μ_2	μ_3	μ_4	μ_5	μ_6

Writing a contrast of the treatment means as

$$c = a_1\mu_1 + a_2\mu_2 + a_3\mu_3 + a_4\mu_4 + a_5\mu_5 + a_6\mu_6$$

Give the coefficients of the contrast you would test to answer each of the following questions. In each case you will test the null hypothesis that the contrast equals zero.

1) Do control mice have the same average lifetimes as laboratory mice?

a_1	a_2	a_3	a_4	a_5	a_6

2) Does reducing calorie intake from 85 to 50 kcal per week increase average life-span?

a_1	a_2	a_3	a_4	a_5	a_6

3) Is there an effect of pre-weaning diet restriction?

a_1	a_2	a_3	a_4	a_5	a_6

4) Does further restriction from 50 to 40 kcal per week further increase average life-span?

a_1	a_2	a_3	a_4	a_5	a_6

5) Does reduction in protein, with the same calories, change average life length?

a_1	a_2	a_3	a_4	a_5	a_6



My Answers

NP	N/N85	N/R50	R/R50	N/R50 lopro	N/R40
μ_1	μ_2	μ_3	μ_4	μ_5	μ_6

Do control mice have the same average lifetimes as laboratory mice?

a_1	a_2	a_3	a_4	a_5	a_6
1	-1	0	0	0	0

Does reducing calorie intake from 85 to 50 kcal per week increase average life-span?

a_1	a_2	a_3	a_4	a_5	a_6
0	1	-1	0	0	0

Is there an effect of pre-weaning diet restriction?

a_1	a_2	a_3	a_4	a_5	a_6
0	0	1	-1	0	0

Does further restriction from 50 to 40 kcal per week further increase average life-span?

a_1	a_2	a_3	a_4	a_5	a_6
0	0	1	0	0	-1

Does reduction in protein, with the same calories, change average life length?

a_1	a_2	a_3	a_4	a_5	a_6
0	0	0	1	-1	0

What's the most promising multiple comparison method? Why?

Data are in an Excel Spreadsheet

Most real data sets these days are Microsoft Excel spreadsheets. The data from this study come in a spreadsheet called `StarvingMice.XLS`. Reading the data from an Excel spreadsheet into SAS is easy with `proc import`. Otherwise, it's a nightmare.



	A	B	C	D	E
1	LIFETIME	DIET			
2	35.5	NP			
3	35.4	NP			
4	34.9	NP			
5	34.8	NP			
6	33.8	NP			
7	33.5	NP			
8	32.6	NP			
9	32.4	NP			
10	31.8	NP			
11	31.6	NP			
12	31.5	NP			
13	31.4	NP			
14	31.4	NP			
15	31.3	NP			
16	30.8	NP			
17	30.7	NP			
18	30.5	NP			
19	30.4	NP			
20	30.2	NP			
21	30.2	NP			
22	30.1	NP			

```
/* hungrymice.sas */
options linesize=79 pagesize=500 noovp formdlim=' ';
title 'Calorie restriction and Longevity in Mice';

/* Read data directly from Excel spreadsheet */
proc import datafile="StarvingMice.XLS" out=mouse1 dbms=xls;
            getnames=yes;
/* Input data file is StarvingMice.XLS
   Output data set is called mouse1
   dbms=xls The input file is an Excel spreadsheet.
            Necessary to read an Excel spreadsheet directly under unix/linux
            Works in PC environment too except for Excel 4.0 spreadsheets
            The xlsx file type is not supported as of SAS Version 9.2
            If there are multiple sheets, use sheet="sheet1" or something.
   getnames=yes   Use column names as variable names                  */

/*  proc print;  */

proc freq;
     tables diet;
proc means;
     class diet;
     var lifetime;

data mouse2;
     set mouse1; /* Now mouse2 is just mouse1, and we can
                    transform the data.  */
     diet0 = diet;
     if diet = 'NP' then diet = 'Feed at Will';
        else if diet = 'lopro' then diet = 'N/R50 lopro';
     label lifetime = 'Life Length in Months';

/* By default, procedures use the most recent SAS data set. If you
   don't want this, use the data= option. */

proc freq;
     title2 'Check Re-labeling of Diet';
     tables diet * diet0 / norow nocol nopercent;

proc glm;
     class diet;
     model lifetime=diet / clparm; /* clparm gives CIs for contrasts down in
                                      the estimate statements. */
     means diet;
     /* Estimate (like Contrast) uses alphabetical order: Be careful!
        Feed at Will | N/N85 | N/R40 | N/R50 | N/R50 lopro | R/R50
        Positive values of contrast mean longer life with less food. */
     estimate 'Feed at Will vs. N/N85'   diet -1  1  0  0  0  0;
     estimate 'N/N85 vs. N/R50'          diet  0 -1  0  1  0  0;
     estimate 'N/R50 vs. R/R50'          diet  0  0  0 -1  0  1;
     estimate 'N/R40 vs. N/R50'          diet  0  0  1 -1  0  0;
     estimate 'N/R50 vs. N/R50 lopro'    diet  0  0  0 -1  1  0;
```

```
% cat hungrymice.lst
```

Calorie restriction and Longevity in Mice 1

The FREQ Procedure

DIET

DIET	Frequency	Percent	Cumulative Frequency	Cumulative Percent
N/N85	57	16.33	57	16.33
N/R40	60	17.19	117	33.52
N/R50	71	20.34	188	53.87
NP	49	14.04	237	67.91
R/R50	56	16.05	293	83.95
lopro	56	16.05	349	100.00

Calorie restriction and Longevity in Mice 2

The MEANS Procedure

Analysis Variable : LIFETIME LIFETIME

DIET	N Obs	N	Mean	Std Dev	Minimum	Maximum
N/N85	57	57	32.6912281	5.1252972	17.9000000	42.3000000
N/R40	60	60	45.1166667	6.7034058	19.6000000	54.6000000
N/R50	71	71	42.2971831	7.7681947	18.6000000	51.9000000
NP	49	49	27.4020408	6.1337010	6.4000000	35.5000000
R/R50	56	56	42.8857143	6.6831519	24.2000000	50.7000000
lopro	56	56	39.6857143	6.9916945	23.4000000	49.7000000

Compare Order

NP	N/N85	N/R50	R/R50	R/R50 lopro	N/R40
μ_1	μ_2	μ_3	μ_4	μ_5	μ_6

The GLM Procedure

Class Level Information

Class	Levels	Values
DIET	6	Feed at Will N/N85 N/R40 N/R50 N/R50 lopro R/R50

Number of Observations Read 349
Number of Observations Used 349

The GLM Procedure

Dependent Variable: LIFETIME Life Length in Months

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	12733.94181	2546.78836	57.10	<.0001
Error	343	15297.41532	44.59888		
Corrected Total	348	28031.35713			

R-Square	Coeff Var	Root MSE	LIFETIME Mean
0.454275	17.21323	6.678239	38.79713

Source	DF	Type I SS	Mean Square	F Value	Pr > F
DIET	5	12733.94181	2546.78836	57.10	<.0001

Source	DF	Type III SS	Mean Square	F Value	Pr > F
DIET	5	12733.94181	2546.78836	57.10	<.0001

The GLM Procedure

Level of DIET	N	LIFETIME Mean	LIFETIME Std Dev
Feed at Will	49	27.4020408	6.13370096
N/N85	57	32.6912281	5.12529723
N/R40	60	45.1166667	6.70340583
N/R50	71	42.2971831	7.76819471
N/R50 lopro	56	39.6857143	6.99169452
R/R50	56	42.8857143	6.68315191

Level of DIET	N	LIFETIME Mean	LIFETIME Std Dev
Feed at Will	49	27.4020408	6.13370096
N/N85	57	32.6912281	5.12529723
N/R40	60	45.1166667	6.70340583
N/R50	71	42.2971831	7.76819471
N/R50 lopro	56	39.6857143	6.99169452
R/R50	56	42.8857143	6.68315191

The GLM Procedure

Dependent Variable: LIFETIME Life Length in Months

Parameter	Estimate	Standard Error	t Value	Pr > |t|
Feed at Will vs. N/N85	5.28918725	1.30100640	4.07	<.0001
N/N85 vs. N/R50	9.60595503	1.18768248	8.09	<.0001
N/R50 vs. R/R50	0.58853119	1.19355007	0.49	0.6223
N/R40 vs. N/R50	2.81948357	1.17109686	2.41	0.0166
N/R50 vs. N/R50 lopro	-2.61146881	1.19355007	-2.19	0.0293

Parameter	95% Confidence Limits	
Feed at Will vs. N/N85	2.73023219	7.84814232
N/N85 vs. N/R50	7.26989727	11.94201279
N/R50 vs. R/R50	-1.75906755	2.93612992
N/R40 vs. N/R50	0.51604814	5.12291900
N/R50 vs. N/R50 lopro	-4.95906755	-0.26387008

Rank-Based Non-parametric Tests

Wilcoxon scores are just ranks of the data: Lowest=1, Highest=n. Tests are based on observed versus expected sums of ranks in each group. Expected sums of ranks are based on the assumption (null hypothesis) that the experimental treatment has no effect, so the data have the same probability distribution within each group. The 2-sample version is called a Wilcoxon test, and the p-sample version is called a Kruskal-Wallis test, or a Kruskal-Wallis one-way analysis of variance by ranks.

```
/* hungrymice2.sas */
options linesize=79 pagesize=500 noovp formdlim=' ' nodate;
title 'Calorie restriction and Longevity in Mice: Nonparametric Analysis';

/* Read data directly from Excel spreadsheet */
proc import datafile="StarvingMice.XLS" out=mouse1 dbms=xls;
            getnames=yes;
data mouse2; set mouse1;
     if diet = 'NP' then diet = 'Feed at Will';
        else if diet = 'lopro' then diet = 'N/R50 lopro';
     label lifetime = 'Life Length in Months';

proc npar1way Wilcoxon;
     title2 'Compare all 6 conditions';
     class diet;
     var lifetime;

/* Pairwise Comparisons */

proc npar1way Wilcoxon;
     title2 'Do control mice have the same average lifetimes';
     title3 'as laboratory mice?';
     where diet = 'Feed at Will' or diet = 'N/N85';
     class diet;
     var lifetime;
proc npar1way Wilcoxon;
     title2 'Does reducing calorie intake from 85 to 50 kcal per week';
     title3'increase average life span?';
     where diet = 'N/N85' or diet = 'N/R50';
     class diet;
     var lifetime;
proc npar1way Wilcoxon;
     title2 'Is there an effect of pre-weaning diet restriction?';
     where diet = 'N/R50' or diet = 'R/R50';
     class diet;
     var lifetime;
proc npar1way Wilcoxon;
     title2 'Does further restriction from 50 to 40  kcal per week further';
     title3 'increase average life span?';
     where diet = 'N/R50' or diet = 'N/R40';
     class diet;
     var lifetime;
proc npar1way Wilcoxon;
     title2 'Does reduction in protein, with the same calories, change';
     title3 'average life length?';
     where diet = 'N/R50' or diet = 'N/R50 lopro';
     class diet;
     var lifetime;
```

Compare all 6 conditions

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
Feed at Will	49	2566.00	8575.0	654.775948	52.367347
N/N85	57	5758.50	9975.0	696.727422	101.026316
N/R50 lopro	56	10078.50	9800.0	691.770239	179.973214
N/R50	71	15334.50	12425.0	758.727147	215.978873
R/R50	56	12396.00	9800.0	691.770239	221.357143
N/R40	60	14941.50	10500.0	711.145714	249.025000

Average scores were used for ties.

Kruskal-Wallis Test

Chi-Square	159.0128
DF	5
Pr > Chi-Square	<.0001

Do control mice have the same average lifetimes
as laboratory mice?

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
Feed at Will	49	1826.50	2621.50	157.795697	37.275510
N/N85	57	3844.50	3049.50	157.795697	67.447368

Average scores were used for ties.

Wilcoxon Two-Sample Test

Statistic	1826.5000
Normal Approximation	
Z	-5.0350
One-Sided Pr < Z	<.0001
Two-Sided Pr > |Z|	<.0001
t Approximation	
One-Sided Pr < Z	<.0001
Two-Sided Pr > |Z|	<.0001

Z includes a continuity correction of 0.5.

Kruskal-Wallis Test

Chi-Square	25.3831
DF	1
Pr > Chi-Square	<.0001

Does reducing calorie intake from 85 to 50 kcal per week
increase average life span?

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
N/N85	57	2271.0	3676.50	208.570468	39.842105
N/R50	71	5985.0	4579.50	208.570468	84.295775

Average scores were used for ties.

Wilcoxon Two-Sample Test

Statistic	2271.0000
Normal Approximation	
Z	-6.7363
One-Sided Pr < Z	<.0001
Two-Sided Pr > |Z|	<.0001
t Approximation	
One-Sided Pr < Z	<.0001
Two-Sided Pr > |Z|	<.0001

Z includes a continuity correction of 0.5.

Kruskal-Wallis Test

Chi-Square	45.4105
DF	1
Pr > Chi-Square	<.0001

Is there an effect of pre-weaning diet restriction?

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
N/R50	71	4512.50	4544.0	205.924023	63.556338
R/R50	56	3615.50	3584.0	205.924023	64.562500

Average scores were used for ties.

Wilcoxon Two-Sample Test

Statistic	3615.5000
Normal Approximation	
Z	0.1505
One-Sided Pr > Z	0.4402
Two-Sided Pr > |Z|	0.8803
t Approximation	
One-Sided Pr > Z	0.4403
Two-Sided Pr > |Z|	0.8806

Z includes a continuity correction of 0.5.

Kruskal-Wallis Test

Chi-Square	0.0234
DF	1
Pr > Chi-Square	0.8784

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
N/R50	71	4238.50	4686.0	216.448595	59.697183
N/R40	60	4407.50	3960.0	216.448595	73.458333

Average scores were used for ties.

Wilcoxon Two-Sample Test

Statistic	4407.5000
Normal Approximation	
Z	2.0652
One-Sided Pr > Z	0.0195
Two-Sided Pr > |Z|	0.0389
t Approximation	
One-Sided Pr > Z	0.0204
Two-Sided Pr > |Z|	0.0409

Z includes a continuity correction of 0.5.

Kruskal-Wallis Test

Chi-Square	4.2744
DF	1
Pr > Chi-Square	0.0387

Does reduction in protein, with the same calories, change average life length?

The NPAR1WAY Procedure

Wilcoxon Scores (Rank Sums) for Variable LIFETIME
Classified by Variable DIET

DIET	N	Sum of Scores	Expected Under H0	Std Dev Under H0	Mean Score
N/R50 lopro	56	3073.0	3584.0	205.926135	54.875000
N/R50	71	5055.0	4544.0	205.926135	71.197183

Average scores were used for ties.

Wilcoxon Two-Sample Test

Statistic	3073.0000
Normal Approximation	
Z	-2.4790
One-Sided Pr < Z	0.0066
Two-Sided Pr > |Z|	0.0132
t Approximation	
One-Sided Pr < Z	0.0072
Two-Sided Pr > |Z|	0.0145

Z includes a continuity correction of 0.5.

Kruskal-Wallis Test

Chi-Square	6.1577
DF	1
Pr > Chi-Square	0.0131

Compare Parametric and Non-parametric results

Overall Test F = 57.10 χ^2 = 159.0128

	t	p	Z	p
Feed at Will vs. N/N85	4.07	<.0001	5.0350	<.0001
N/N85 vs. N/R50	8.09	<.0001	6.7363	<.0001
N/R50 vs. R/R50	0.49	0.6223	0.1505	0.8803
N/R40 vs. N/R50	2.41	0.0166	2.0652	0.0389
N/R50 vs. N/R50 lopro	-2.19	0.0293	-2.4790	0.0145

Note: The signs of some Z statistics were changed manually.